
Autor:

4. Februar 2008Datum:

Veit Wahlich

The use of a camel image in association with Perl is a trademark of O'Reilly Media, Inc. Used with permission.

Perl Mongers im Ruhrgebiet

Ruhr.pm

Template Version 0.1

EMail:

http://ruhr.pm.org/

Bildverarbeitung mit PerlMagick
Eine praktische Einfuehrung in Image::Magick

veit AT ruhr.pm.org



Dieses Dokument wurde veröffentlicht unter der Lizenz

Creative Commons Attribution-Noncommercial-NoDerivs 2.0 Germany

Die Lizenz sowie entsprechende Übersetzungen sind einsehbar unter:
http://creativecommons.org/licenses/by-nc-nd/2.0/de/

Zusammenfassend ergeben sich hieraus die folgenden Rechte:

Sie dürfen das Werk vervielfältigen, verbreiten und öffentlich zugänglich machen.

Diese Rechte werden Ihnen unter den folgenden Bedingungen gewährt:

Namensnennung. Sie müssen den Namen des Autors/Rechteinhabers in der von ihm festgelegten
Weise nennen (wodurch aber nicht der Eindruck entstehen darf, Sie oder die Nutzung des Werkes
durch Sie würden entlohnt).

Keine kommerzielle Nutzung. Dieses Werk darf nicht für kommerzielle Zwecke verwendet werden.

Keine Bearbeitung. Dieses Werk darf nicht bearbeitet oder in anderer Weise verändert werden.

Im Falle einer Verbreitung müssen Sie anderen die Lizenzbedingungen, unter welche dieses Werk fällt,
mitteilen.
 
Jede der vorgenannten Bedingungen kann aufgehoben werden, sofern Sie die Einwilligung des Rechteinhabers 
dazu erhalten.

Diese Lizenz lässt die Urheberpersönlichkeitsrechte unberührt. 
Diese Übersichtseite wurde erstellt von Veit Wahlich und basiert auf dem Lizenztext

"Attribution-Noncommercial-NoDerivs 2.0 Germany" der CCi Germany, http://creativecommons.org/international/de/.



Seite 2

Perl Mongers im Ruhrgebiet

Ruhr.pm

Veit Wahlich 4. Februar 2008

Was ist Image::Magick?

● Binding, das den enormen Funktionsumfang von 
ImageMagick vollstaendig abbildet
– optimal fuer die Verarbeitung von Grafiken und Fotos

● Transformation, Komposition, Beschriftung, Filter
● Optimierung des Bildinhalts

– Farben, Gamma, Kontrast, Helligkeit, Schaerfe, ...
● Optimierung der Dateigroesse

– Farbtiefe, Dateiformat, Kompressionsgrad, ... 
– aber auch geeignet fuer die Erstellung von Grafiken

● Funktionen fuer das Zeichnen geometrischer Formen

Was ist Image::Magick?

● vollstaendig objektorientiertes API
– Schreibweise der Methoden allerdings nicht Perl

ueblich

– unterstuetzt neben einfachen Bildern auch 
ebenenaehnliche Behandlung von Unterbildern

● kann auch verwendet werden, um Animationen zu 
erstellen

Was kann Image::Magick?

● dieser Vortrag kann nur wenige dieser Methoden 
behandeln und deren Parameter nicht 
vollstaendig abdecken

● Dokumentation aller Manipulationsmethoden 
inkl. aller moeglichen Parameter:
– http://www.imagemagick.org/script/perlmagick.php 

#manipulate

Beispiele

● die Funktionsweise von Image::Magick soll hier 
mit einigen praxisnahen Beispielen demonstriert 
werden:
– KontaktabzugIndexprint

– TVOptimierung

– PsychoLama!

Kontaktabzug-Indexprint

● erzeugt einen Indexprint, wie er von 
Kontaktabzuegen in der Fotoentwicklung 
bekannt ist

● stellt wichtige Image::MagickMethoden vor
– Read, Montage, Write, Display

● beschreibt interessante PerlFunktionen, die evtl. 
nicht jeder regelmaessig verwendet
– glob, map (VoidKontext)

Kontaktabzug-Indexprint

use strict;
use warnings;
use Image::Magick;

# Dateimaske fuer EingabeBilder.
my $maske   = 'bilder/*.jpg';

# Dateiname fuer die Ausgabedatei.
my $index   = 'indexprint.jpg';

# Max. Groesse der Thumbnails.
my $groesse = '120x120';

# Anzahl Bilder pro Zeile.
my $spalten = 4;

# Lese anhand der Dateimaske die Dateinamen
# der Bilder ein.
 
my @bilder = glob($maske);

# Jede Zeile soll $spalten Bilder enthalten,
# bestimme die Anzahl der Zeilen. 

my $zeilen = int(@bilder / $spalten)
             + (@bilder % $spalten ? 1 : 0);

ARRAY = glob STRING

glob liest alle (nicht versteckten)
Verzeichniseintraege, die auf die
Maske in STRING passen, und
speichert sie in ARRAY.

Berechnet den ganzzahligen Anteil
des Quotienten aus der Anzahl der
Bilder durch die Anzahl der Bilder
pro Spalte.
Addiere ausserdem 1 hinzu, falls
die Ganzzahldivision der Anzahl
der Bilder durch die Anzahl pro
Spalte einen Restwert ergibt.

# Erzeuge in $i ein neues Image::Magick
# Objekt. 

my $i = new Image::Magick;

# Lese alle Bilder als Ebenen in $i ein.

$i>Read(@bilder);

STRING = $i>Read ARRAY

Die Methode Read liest alle
Grafiken aus ARRAY als neue
Ebenen in das Image::Magick
Objekt $i ein.
In STRING werden eventuelle
Fehlermeldungen gespeichert.

Hier werden alle Dateien aus dem
Array @bilder eingeladen,
eventuelle Fehler werden ignoriert.

Kontaktabzug-Indexprint

# Entferne alle Pfadangaben aus den
# Dateinamen. 

map{s|^.*/||}@bilder;

# Durchlaufe alle Bilder und gebe jedem
# Bild seinen Dateinamen als Label. 

foreach my $j (@{$i}){

  $j>Label(shift(@bilder));

}

map BLOCK ARRAY

map im VoidKontext verarbeitet
jedes Element aus ARRAY ueber
$_ in BLOCK und ersetzt das
Element in ARRAY durch den
Rueckgabewert.

Hier wird $_ durch einen Regexp
modifiziert, der alles vom Zeilen
anfang bis zum letzten Slash (“/”)
durch einen leeren String ersetzt.

# Erstelle aus den Ebenen in $i einen
# Indexprint in einem neuen Image::Magick
# Objekt $k. 

my $k = $i>Montage(

    # Ueberschrift fuer den Indexprint. 
    title      => $maske,

    # Anordnung der Einzelbilder. 
    tile       => $spalten.'x'.$zeilen,

    # Max. Groesse und Abstand der Bilder. 
    geometry   => $groesse.'+10+10',

    # Einen 2px breiten Rand um die Bilder. 
    border     => 2,

    # Schatten fuer die Bilder. 
    shadow     => 1,

  );

$k = $i>Montage HASH

Die Methode Montage erzeugt aus
den Ebenen eines Image::Magick
Objekts ein neues Objekt mit einer
Tabelle von Thumbnails.

Montage kennt viele Parameter in
Form von HASH, die wichtigsten,
title, tile, geometry, border
und shadow werden hier
angewendet, das Beispielscript
indexprint.pl beschreibt noch einige
weitere.

# Schreibe den Indexprint in eine
# Grafikdatei. 

$k>Write(

    # Dateiname fuer die Ausgabedatei.
    filename => $index,

    # Qualitaets/Kompressionsstufe.
    quality  => 90

  );

STRING = $k>Write HASH

Die Methode Write schreibt den
Inhalt von $k als Grafikdatei auf den
Datentraeger oder auf ein Handle,
STRING speichert eventuelle Fehler.

Write kann mit vielen Parameter in
Form von HASH aufgerufen werden,
hier werden mit filename ein
Dateiname und mit quality ein
Kompressionsgrad angegeben.
Das Grafikformat der Ausgabe wird
wahlweise durch die Dateiendung
oder ein Praefix wie “png:” definiert.

# Zeige den Indexprint auf dem Bildschirm
# an. 

$k>Display();

STRING = $k>Display

Mit der Methode Display wird ein
Fenster mit dem (ersten) Bild aus
dem Image::MagickObjekt $k auf
dem Bildschirm angezeigt.
STRING kann eventuelle Fehler
meldungen enthalten.

Das Programm wird erst fortgesetzt,
wenn das Fenster geschlossen
wurde.

TV-Optimierung

● erzeugt aus Bildern in einem Verzeichnis 
skalierte Versionen, die einen schwarzen 
Mindestrand besitzen, damit bei der Ausgabe auf 
Fernsehern (ueber DVDPlayer o.ae.) keine 
Bildteile abgeschnitten werden

● zeigt weitere wichtige Image::MagickMethoden
– Set, Scale, Composite

● zeigt eine weitere interessante PerlFunktion
– grep

TV-Optimierung

use strict;
use warnings;
use Image::Magick;

# Verzeichnis mit EingabeBildern. 
my $dir_in           = 'input';

# Verzwichnis fuer Ausgabe. 
my $dir_out          = 'output';

# Groesse der Ausgabebilder. 
my $groesse_leinwand = '800x600';

# Groesse der eingebetteten Bilder. 
my $groesse_bild     = '768x576';

# Ausgabequalitaet/kompression.
my $qualitaet        = 85;

TV-Optimierung

# Oeffne das Eingabeverzeichnis in $dh.

opendir(my $dh, $dir_in);

# Lese alle Eintraege im VerzeichnisHandle
# $dh und filtere unpassende Eintraege aus.

my @bilder = grep{

    # Nur Eintraege, die Dateien sind...
    f $dir_in.'/'.$_

    # ... und deren Name nur aus alphanum.
    # Zeichen besteht und auf .jpg, .jpeg
    # oder .png endet (case insensitive).
    && /^\w+\.(?:jpe?g|png)$/i

  }readdir($dh);

# Schliesse das VerzeichnisHandle $dh.

closedir($dh);

ARRAY2 = grep BLOCK ARRAY1

grep im ArrayKontext verarbeitet
jedes Element aus ARRAY1 und
speichert nur die Elemente in
ARRAY2, fuer die BLOCK mit der
Laufvariable $_ zu true evaluiert.

TV-Optimierung

# Durchlaufe alle Bilder und verarbeite sie. 

foreach my $datei (@bilder){

  printf("Verarbeite %s...\n", $datei);

  # Erzeuge ein Image::MagickObjekt $i.

  my $i = new Image::Magick;

  # $i, setze die Leinwandgroesse und 

  $i>Set(size => $groesse_leinwand);

  # Lese eine schwarze Flaeche als Bild ein. 

  $i>Read('xc:black');

$i>Set HASH

Die Methode Set setzt im
Image::MagickObjekt $i die Werte
der Eigenschaften in HASH.

Hier wird die Groesse des Bildes
bzw. der Leinwand definiert.

Read erzeugt mit einem Pseudo
dateinamen aus dem Prefix “xc:”
und einer Farbdefinition ein Bild mit
einer farbigen Flaeche in Bild
groesse.

TV-Optimierung

  # Erzeuge ein weiteres Image::Magick
  # Objekt $j.

  my $j = new Image::Magick;

  # Lade lade die zu verarbeitende Datei
  # aus dem Eingabeverzeichnis hinein.

  $j>Read($dir_in.'/'.$datei);

  # Das Bild wird auf die maximal darstell
  # bare Groesse des Fernsehers skaliert.

  $j>Scale(geometry => $groesse_bild);

$j>Scale HASH

Die Methode Scale skaliert das
Image::MagickObjekt $j mit den in
HASH uebergebenen Parametern
auf eine neue Groesse.

HASH enthaelt wahlweise ein Hash
Element mit dem Schluessel
geometry oder mindestens eines
der Elemente width und height.

Entgegen der Doku sind geometry
und width/height nicht aequiva
lent, da nur geometry das Seiten
verhaeltnis beruecksichtigt.

  # Lege das Bild $j zentriert ueber die
  # schwarze Leinwand in $i. 

  $i>Composite(

      # Das einzufuegende Bild.
      image   => $j,

      # Position von $j auf $i zentrieren.
      gravity => 'Center'

    );

  # Schreibe das zusammengefuegte Bild in
  # das AusgabeVerzeichnis. 

  $i>Write(
      filename => $dir_out.'/'.$datei,
      quality  => $qualitaet
    );

} # Ende foreach $datei (@bilder)

$i>Composite HASH

Die Methode Composite legt das
in HASH unter dem Schluessel
image uebergebene Image::Magick
Objekt ueber das Bild im Objekt $i.

Weitere wichtige Parameter in HASH
koennen Angaben zur Position mit x
und y sein, oder, wie hier, die Posi
tionierung per gravity.

Es koennen aber auch komplexe
weitere Eigenschaften definiert
werden, wie z.B. betroffene Farb
kanaele oder Rotation.



Seite 25

Perl Mongers im Ruhrgebiet

Ruhr.pm

Veit Wahlich 4. Februar 2008

Psycho-Lama!

● generiert aus fiesen Bildern noch fiesere 
Animationen

● stellt wichtige Image::MagickMethoden vor
– Get, Rotate, Scale (alternative Anwendung), Negate, 

Crop, Write (alternative Anwendung), Animate

● beschreibt PerlFunktionen, die evtl. nicht jedem 
bekannt sind
– map (ArrayKontext)

Psycho-Lama!

use strict;
use warnings;
use Image::Magick;

# Festlegen der Parameter fuer die
# Verarbeitung.

my $conf = {

    # Dateiname der EingabeGrafik.
    input  => 'lama.png',

    # Dateiname der auszugebenden Animation.
    output => 'lama.gif',

    # Einzelne Ebenen um 2.5° drehen...
    rotate => 2.5,

    # ... und um 10% vergroessern.
    zoom   => 1.10,

    # Zwischen den Bildwechseln 50ms Pause.
    delay  => 5

  };

Psycho-Lama!

# Ein neues Image::MagickObjekt erzeugen.

my $i = new Image::Magick;

# Die EingabeGrafik wird 4x geladen.

$i>Read(map{$conf>{input}}(0..3));

ARRAY2 = map BLOCK ARRAY1

map im ArrayKontext verarbeitet
jedes Element aus ARRAY1 ueber
$_ in BLOCK und speichert den
Rueckgabewert in ARRAY2.

Hier wird fuer jedes Element aus
dem Array (0,1,2,3) lediglich
der Inhalt von $conf>{input}
zurueck gegeben.

Interpolierter Aufruf:
$i>Read('lama.png',
'lama.png','lama.png',
'lama.png');

Psycho-Lama!

# Als Referenzwert fuer die spaetere
# Verarbeitung werden Breite und Hoehe der
# Ebene 0 gespeichert.

my($w,$h) = $i>[0]>Get('width','height');

ARRAY2 = $i>Get ARRAY1

Die Methode Get bezieht die Werte
der Eigenschaften ARRAY1 aus
dem Image::MagickObjekt $i bzw.
hier aus der ersten Ebene
($i>[0]) des Objekts und
speichert sie in ARRAY2.

Hier wird die Breite und Hoehe der
ersten Ebene von $i ausgelesen
und in $w und $h gespeichert.

# Ebene 1 und 3 werden erst rotiert und
# dann hochskaliert.

for(1,3){

  $i>[$_]>Rotate(
      degrees => $conf>{rotate}
    );

  $i>[$_]>Scale(
      width  => $w*$conf>{zoom},
      height => $h*$conf>{zoom}
    );

}

3 3

3 3

Psycho-Lama!

# Ebene 2 und 3 in ein Negativ verwandeln.

for(2,3){

  $i>[$_]>Negate();

}
2 2

33

Psycho-Lama!

# Berechne das Delta zwischen Original
# groesse (Ebene 0) und neuer Groesse nach
# Rotation und Hochskalierung (Ebenen 1, 3).

my($dx,$dy)= $i>[1]>Get('width','height');
$dx = $w;
$dy = $h;

0

$w

$
h

1

$dx

$
d
y



Seite 33

Perl Mongers im Ruhrgebiet

Ruhr.pm

Veit Wahlich 4. Februar 2008

Psycho-Lama!

# Beschneide die Ebenen 1 und 3 auf
# Originalgroesse, halbiere die Deltas,
# um den Abstand zum oberen linken Rand zu
# bestimmen.

for(1,3){

  $i>[$_]>Crop(
      width  => $w,
      height => $h,
      x      => $dx/2,
      y      => $dy/2
    );

}

1

$w

$
h

$dx/2$
d
y
/
2

1

Psycho-Lama!

# Setze die virtuelle Position aller Ebenen
# auf die absolute Position 0,0.
# Setze ausserdem die Zeitspanne zwischen
# den Bildwechseln der Ebenen in der
# Animation (Wert*10ms).

for(0..3){

  $i>[$_]>Set(
      page  => '+0+0',
      delay => $conf>{delay}
    );

}

1

1

Psycho-Lama!

# Schreibe die fertige Animation in eine
# Datei...

$i>Write($conf>{output});

# ... und zeige sie auf dem Bildschirm an.

$i>Animate();

STRING2 = $i>Write STRING1

Hier wird die Methode Write mit
einer einfacheren Syntax verwendet,
in der nur der Dateiname der
Ausgabedatei, STRING1,
uebergeben wird. STRING2 enthaelt
eventuelle Fehlermeldungen.

STRING = $i>Animate

Animate ist das Aequivalent zu
Display fuer das Anzeigen der
Ebenen als Animation statt nur der
ersten Ebene als Bild.

Vielen Dank
für Eure Aufmerksamkeit

Links

● PerlMagick (inkl. Image::Magick) und 
Dokumentation
– http://www.imagemagick.org/script/perlmagick.php



#!/usr/bin/perl
#
# indexprint.pl
# Erzeugt einen Indexprint, wie er z.B. von Kontaktabzuegen in der
# Fotoentwicklung bekannt ist.
#
# Ein Beispiel zum Vortrag "Bildverarbeitung mit PerlMagick" der Ruhr.pm,
# gehalten von Veit Wahlich, (C)2008.
#

use strict;
use warnings;
use Image::Magick;

my $maske   = 'bilder/*.jpg';           # Dateimaske fuer Eingabe-Bilder.
my $index   = 'indexprint.jpg';         # Dateiname fuer die Ausgabedatei.
my $groesse = '120x120';                # Max. Groesse der Thumbnails.
my $spalten = 4;                        # Anzahl Bilder pro Zeile.

# Lese anhand der Dateimaske die Dateinamen der Bilder ein.
my @bilder = glob($maske);

# Jede Zeile soll $spalten Bilder enthalten, bestimme die Anzahl der Zeilen.
# Berechne Ganzzahl des Quotienten aus der Anzahl der Bilder durch die Anzahl
# der Spalten und addiere 1, falls die Division einen Rest enthaelt.
my $zeilen = int(@bilder / $spalten) + (@bilder % $spalten ? 1 : 0);

# Erzeuge in $i ein neues Image::Magick-Objekt.
my $i = new Image::Magick;

# Lese alle Bilder als Ebenen in $i ein.
$i->Read(@bilder);

# Entferne alle Pfadangaben aus den Dateinamen.
map{s|^.*/||}@bilder;

# Durchlaufe alle Bilder und gebe jedem Bild seinen Dateinamen als Label.
foreach my $j (@{$i}){
	$j->Label(shift(@bilder));
}

# Erstelle aus den Ebenen in $i einen Indexprint in einem neuen
# Image::Magick-Objekt $k.
my $k = $i->Montage(
		title      => $maske,               # Ueberschrift fuer den Indexprint.
		tile       => $spalten.'x'.$zeilen, # Anordnung der Einzelbilder.
		geometry   => $groesse.'+10+10',    # Max. Groesse und Abstand der Bilder.
		border     => 2,                    # Einen 2px breiten Rand um die Bilder.
		shadow     => 1,                    # Schatten fuer die Bilder.
		background => '#ffffff',            # Hintergrundfarbe fuer den Indexprint.
		fill       => '#000000',            # Vordergrundfarbe fuer Beschriftung.
		font       => 'Nimbus Sans L',      # Scrhift fuer die Beschriftung.
		pointsize  => 14                    # Referenz-Schriftgroesse.
	);

# Schreibe den Indexprint in eine Grafikdatei.
$k->Write(
		filename => $index,                 # Dateiname der Ausgabegrafik.
		quality  => 90                      # Qualitaets-/Kompressionsstufe.
	);

# Zeige den Indexprint auf dem Bildschirm an.
$k->Display();



#!/usr/bin/perl
# 
# psycho-lama.pl
# Generiert aus fiesen Bildern noch fiesere Animationen.
# 
# Ein Beispiel zum Vortrag "Bildverarbeitung mit PerlMagick" der Ruhr.pm, 
# gehalten von Veit Wahlich, (C)2008.
# 

use strict;
use warnings;
use Image::Magick;

my $conf = {
		input  => 'lama.png',               # Mache aus dem Lama...
		output => 'psycho-lama.gif',        # ... ein fieses Psycho-Lama!
		rotate => -2.5,                     # Bild um -2.5° drehen...
		zoom   => 1.10,                     # ... und um 10% vergroessern.
		delay  => 5                         # 50ms Pause zwischen den Bildern.
	};

my $i = new Image::Magick;

# 4x das Eingabe-Bild in Ebenen laden.
$i->Read(map{$conf->{input}}(0..3));

# Breite und Hoehe der Ebene 0 als Referenz bestimmen.
my($w,$h) = $i->[0]->Get('width','height');

# Das Bild auf Ebene 0, $i->[0], bleibt unveraendert, aber...

# ... die Ebenen 1 und 3 werden etwas gedreht und vergroessert, ...
for(1,3){
	$i->[$_]->Rotate(degrees => $conf->{rotate});
	$i->[$_]->Scale(
			width  => $w*$conf->{zoom},
			height => $h*$conf->{zoom}
		);
}

# ... und die Ebenen 2 und 3 in ein Negativ verwandelt.
for(2,3){
	$i->[$_]->Negate();
}

# Die Ebene 1 und 3 muessen nach der Vergroesserung und Rotation wieder
# auf normale Groesse gebracht werden.

# Bestimme dafuer die Groessen-Differenz zwischen der Groesse der Ebene 1
# und der Referenzgroesse.
my($dx,$dy) = $i->[1]->Get('width','height');
$dx -= $w;
$dy -= $h;

# Beschneide die Ebenen 1 und 3 auf die Groesse des Originalbildes,
# die linke obere Ecke entspricht dabei der halben Groessen-Differenz.
for(1,3){
	$i->[$_]->Crop(
			width  => $w,
			height => $h,
			x      => $dx/2,
			y      => $dy/2
		);
}

# Die virtuelle linke obere Ecke aller Ebenen wird in die absolute obere linke
# Ecke gesetzt und die Zeitspanne zwischen den Bildwechseln wird gesetzt.
for(0..3){
	$i->[$_]->Set(
			page  => '+0+0',
			delay => $conf->{delay}
		);
}

# Nun muss nur noch die fertige Animation gespeichert werden.
$i->Write($conf->{output});

# Schliesslich wird die fertige Animation noch auf dem Bildschirm angezeigt.
$i->Animate();


#!/usr/bin/perl
#
# tv-opt.pl
# Erzeugt aus Bildern im Eingabeverzeichnis skalierte Bilder im Ausgabe-
# verzeichnis, die einen schwarzen Mindestrand besitzen, damit bei der Ausgabe
# auf Fernsehern (ueber DVD-Player o.ae.) keine Bildteile abgeschnitten werden.
#
# Ein Beispiel zum Vortrag "Bildverarbeitung mit PerlMagick" der Ruhr.pm,
# gehalten von Veit Wahlich, (C)2008.
#

use strict;
use warnings;
use Image::Magick;

my $dir_in           = 'input';         # Verzeichnis mit Eingabe-Bildern.
my $dir_out          = 'output';        # Verzwichnis fuer Ausgabe.
my $groesse_leinwand = '800x600';       # Groesse der Ausgabebilder.
my $groesse_bild     = '768x576';       # Groesse der eingebetteten Bilder.
my $qualitaet        = 85;              # Ausgabequalitaet/-kompression.

# Lese alle Eintraege aus dem Eingabeverzeichnis und filtere dabei nur
# normale Dateien heraus, die ausserdem nur alphanumerische Zeichen (plus "_")
# enthalten und die Endung .jpg, .jpeg oder .png tragen.
opendir(my $dh, $dir_in);
my @bilder = grep{-f $dir_in.'/'.$_ && /^\w+\.(?:jpe?g|png)$/i}readdir($dh);
closedir($dh);

# Durchlaufe alle Bilder und verarbeite sie.
foreach my $datei (@bilder){

	printf("Verarbeite %s...\n", $datei);

	# Erzeuge ein neues Image::Magick-Objekt $i, setze die Leinwandgroesse und
	# "lese" eine schwarze Flaeche hinein.
	my $i = new Image::Magick;
	$i->Set(size => $groesse_leinwand);
	$i->Read('xc:black');

	# Erzeuge ein weiteres Image::Magick-Objekt $j, lade die Eingabedatei hinein
	# und skaliere das Bild auf die maximale Bildgroesse herunter.
	my $j = new Image::Magick;
	$j->Read($dir_in.'/'.$datei);
	$j->Scale(geometry => $groesse_bild);

	# Lege das Bild $j zentriert ueber die schwarze Leinwand in $i.
	$i->Composite(
			image   => $j,
			gravity => 'Center'
		);

	# Schreibe das zusammengefuegte Bild in das Ausgabe-Verzeichnis.
	$i->Write(
			filename => $dir_out.'/'.$datei,
			quality  => $qualitaet
		);

}


